```
> #  Geometric constructibility of 3-fans
  #  (addendum to D. Ahmed, G. Czédli, and E. K. Horváth's paper)
  #  With Maple V Release 5, on October 19, 2017
> #     Start: 19:30
> #
> #                  First way to chek Proposition 3.4:
  #
  #
> h1show:= cos(alpha+beta)^2 - cos(delta-gamma)^2; # This = 0
```

$$h1show := \cos(\alpha+\beta)^2 - \cos(\delta-\gamma)^2$$

```
> h1:=expand(h1show);

  # Instead of delta', we simply write delta
```

$$h1 := \cos(\alpha)^2\cos(\beta)^2 - 2\cos(\alpha)\cos(\beta)\sin(\alpha)\sin(\beta) + \sin(\alpha)^2\sin(\beta)^2 - \cos(\delta)^2\cos(\gamma)^2 - 2\cos(\delta)\cos(\gamma)\sin(\delta)\sin(\gamma) - \sin(\delta)^2\sin(\gamma)^2$$

```
> s1:=2*cos(alpha)*cos(beta)*sin(alpha)*sin(beta)+2*cos(delta)*cos
  (gamma)*sin(delta)*sin(gamma);
```

$$s1 := 2\cos(\alpha)\cos(\beta)\sin(\alpha)\sin(\beta) + 2\cos(\delta)\cos(\gamma)\sin(\delta)\sin(\gamma)$$

```
> h1+s1;
```

$$\cos(\alpha)^2\cos(\beta)^2 + \sin(\alpha)^2\sin(\beta)^2 - \cos(\delta)^2\cos(\gamma)^2 - \sin(\delta)^2\sin(\gamma)^2$$

```
> h2show:= (h1+s1)^2 - s1^2; # this=0 since h1=0
```

$$h2show := (\cos(\alpha)^2\cos(\beta)^2 + \sin(\alpha)^2\sin(\beta)^2 - \cos(\delta)^2\cos(\gamma)^2 - \sin(\delta)^2\sin(\gamma)^2)^2 - (2\cos(\alpha)\cos(\beta)\sin(\alpha)\sin(\beta) + 2\cos(\delta)\cos(\gamma)\sin(\delta)\sin(\gamma))^2$$

```
> h2:=expand(h2show); # =0
```

$$h2 := \cos(\alpha)^4\cos(\beta)^4 - 2\cos(\alpha)^2\cos(\beta)^2\sin(\alpha)^2\sin(\beta)^2 - 2\cos(\alpha)^2\cos(\beta)^2\cos(\delta)^2\cos(\gamma)^2 - 2\cos(\alpha)^2\cos(\beta)^2\sin(\delta)^2\sin(\gamma)^2 + \sin(\alpha)^4\sin(\beta)^4 - 2\sin(\alpha)^2\sin(\beta)^2\cos(\delta)^2\cos(\gamma)^2 - 2\sin(\alpha)^2\sin(\beta)^2\sin(\delta)^2\sin(\gamma)^2 + \cos(\delta)^4\cos(\gamma)^4 - 2\cos(\delta)^2\cos(\gamma)^2\sin(\delta)^2\sin(\gamma)^2 + \sin(\delta)^4\sin(\gamma)^4 - 8\cos(\alpha)\cos(\beta)\sin(\alpha)\sin(\beta)\cos(\delta)\cos(\gamma)\sin(\delta)\sin(\gamma)$$

```
> s2:=
  8*cos(alpha)*cos(beta)*sin(alpha)*sin(beta)*cos(delta)*cos(gamma
  )*sin(delta)*sin(gamma);
```

$$s2 := 8\cos(\alpha)\cos(\beta)\sin(\alpha)\sin(\beta)\cos(\delta)\cos(\gamma)\sin(\delta)\sin(\gamma)$$

```
> check:=expand(h2+s2-(h1+s1)^2); # Not needed in the future
```

$$check := -4\cos(\alpha)^2\cos(\beta)^2\sin(\alpha)^2\sin(\beta)^2 - 4\cos(\delta)^2\cos(\gamma)^2\sin(\delta)^2\sin(\gamma)^2$$

```
> h3:=(h2+s2)^2 - s2^2; # surely=0 since h2=0
```

$$h3 := (\cos(\alpha)^4\cos(\beta)^4 - 2\cos(\alpha)^2\cos(\beta)^2\sin(\alpha)^2\sin(\beta)^2 - 2\cos(\alpha)^2\cos(\beta)^2\cos(\delta)^2\cos(\gamma)^2 - 2\cos(\alpha)^2\cos(\beta)^2\sin(\delta)^2\sin(\gamma)^2 + \sin(\alpha)^4\sin(\beta)^4 - 2\sin(\alpha)^2\sin(\beta)^2\cos(\delta)^2\cos(\gamma)^2 - 2\sin(\alpha)^2\sin(\beta)^2\sin(\delta)^2\sin(\gamma)^2 + \cos(\delta)^4\cos(\gamma)^4 - 2\cos(\delta)^2\cos(\gamma)^2\sin(\delta)^2\sin(\gamma)^2 + \sin(\delta)^4\sin(\gamma)^4)^2$$

$- 64 \cos(\alpha)^2 \cos(\beta)^2 \sin(\alpha)^2 \sin(\beta)^2 \cos(\delta)^2 \cos(\gamma)^2 \sin(\delta)^2 \sin(\gamma)^2$

```
> h4:=subs(sin(alpha)=sqrt(1-cos(alpha)^2),
           sin(beta)=sqrt(1-cos(beta)^2),
           sin(gamma)=sqrt(1-cos(gamma)^2),
           sin(delta)=sqrt(1-cos(delta)^2),h3);
```

$$h4 := (\cos(\alpha)^4 \cos(\beta)^4 - 2\cos(\alpha)^2 \cos(\beta)^2 (1-\cos(\alpha)^2)(1-\cos(\beta)^2) - 2\cos(\alpha)^2\cos(\beta)^2\cos(\delta)^2\cos(\gamma)^2 - 2\cos(\alpha)^2\cos(\beta)^2(1-\cos(\delta)^2)(1-\cos(\gamma)^2) + (1-\cos(\alpha)^2)^2(1-\cos(\beta)^2)^2 - 2(1-\cos(\alpha)^2)(1-\cos(\beta)^2)\cos(\delta)^2\cos(\gamma)^2 - 2(1-\cos(\alpha)^2)(1-\cos(\beta)^2)(1-\cos(\delta)^2)(1-\cos(\gamma)^2) + \cos(\delta)^4\cos(\gamma)^4 - 2\cos(\delta)^2\cos(\gamma)^2(1-\cos(\delta)^2)(1-\cos(\gamma)^2) + (1-\cos(\delta)^2)^2(1-\cos(\gamma)^2)^2)^2 - 64 \cos(\alpha)^2\cos(\beta)^2(1-\cos(\alpha)^2)(1-\cos(\beta)^2)\cos(\delta)^2\cos(\gamma)^2(1-\cos(\delta)^2)(1-\cos(\gamma)^2)$$

```
> h5:=subs( cos(alpha)=a*u, cos(beta)=b*u,
            cos(gamma)=u, cos(delta)=p, h4);
```

$$h5 := (a^4 u^8 b^4 - 2a^2u^4b^2(1-a^2u^2)(1-b^2u^2) - 2a^2u^6b^2p^2 - 2a^2u^4b^2(1-p^2)(1-u^2) + (1-a^2u^2)^2(1-b^2u^2)^2 - 2(1-a^2u^2)(1-b^2u^2)p^2u^2 - 2(1-a^2u^2)(1-b^2u^2)(1-p^2)(1-u^2) + p^4u^4 - 2p^2u^2(1-p^2)(1-u^2) + (1-p^2)^2(1-u^2)^2)^2 - 64a^2u^6b^2(1-a^2u^2)(1-b^2u^2)p^2(1-p^2)(1-u^2)$$

```
> hshow:=subs(u=sqrt(x), h5);
```

$$hshow := (a^4x^4b^4 - 2a^2x^2b^2(1-a^2x)(1-b^2x) - 2a^2x^3b^2p^2 - 2a^2x^2b^2(1-p^2)(1-x) + (1-a^2x)^2(1-b^2x)^2 - 2(1-a^2x)(1-b^2x)p^2x - 2(1-a^2x)(1-b^2x)(1-p^2)(1-x) + p^4x^2 - 2p^2x(1-p^2)(1-x) + (1-p^2)^2(1-x)^2)^2 - 64a^2x^3b^2(1-a^2x)(1-b^2x)p^2(1-p^2)(1-x)$$

```
> h:=sort(collect(simplify(expand(hshow)),x),x);
```

$$h := 16a^4b^4x^6 + (-16a^2b^6p^2 - 16a^4b^4 - 16a^6b^2p^2 - 16a^4b^2 - 16a^2b^4 - 16a^2b^2p^2 + 8a^2b^2 + 8a^2b^6 + 8a^6b^2)x^5 + (1 + 24a^2p^2b^4 + 24a^4p^2b^2 - 16a^4b^4p^2 + 8a^2b^6p^2 + 16a^4b^4p^4 + 8a^6b^2p^2 + 8a^6p^2 - 16a^4p^2 + 8b^2p^2 + 6a^4 - 4a^2 + 6b^4 + 6a^4b^4 + a^8 - 4b^6 - 4b^2 - 4a^6 + 24a^2b^2p^2 + 8a^2p^2 + 4a^4b^2 + 4a^2b^4 - 4a^2b^6 - 4a^6b^2 + 4a^2b^2 - 16b^4p^2 + 8b^6p^2 + b^8 + 16b^4p^4 + 16a^4p^4)x^4 + (-16a^4p^4 + 4a^2p^2 + 4a^4p^2 - 4a^6p^2 + 4a^2p^2b^4 - 40a^2b^2p^2 - 16a^2p^4 - 16b^4p^4 - 16a^4p^4b^2 - 16b^4p^4a^2 + 4a^4p^2b^2 - 16b^2p^4 - 4p^2 - 16p^6a^2b^2 + 4b^4p^2 - 4b^6p^2 + 4b^2p^2 + 24a^2p^4b^2)x^3 + (8p^6a^2b^2 + 6a^4p^4 + 4b^2p^4 + 8a^2p^6 + 6b^4p^4 + 8b^2p^6 + 4a^2p^4 + 6p^4 + 4a^2p^4b^2)x^2 + (-4a^2p^6 - 4p^6 - 4b^2p^6)x + p^8$$

```
> irreduc(h); # h is of degree 6; if h is irreducible, then the
  3-fan is not constructible
```

true

```
> # For the paper, a less complex derivative of h will do:
```

```
> hsubs1:=subs(a=2,b=3,p=2,h); irreduc(hsubs1);
```

$$hsubs1 := 20736\,x^6 - 225792\,x^5 + 453376\,x^4 - 180224\,x^3 + 37632\,x^2 - 3584\,x + 256$$

$$true$$

```
> # Let us see some particular experiments:
> hsub:=subs(p=-2/3,h): irreduc(hsub); # A third of a circle,
c.angle=2Pi/3
```

$$true$$

```
> hsub:=subs(p=0,h): irreduc(hsub); # The Thalesian case
```

$$false$$

```
> factor(hsub);
# This shows that the 3-gon is constructible if the central
angle=Pi
```

$$x^4\,(4\,a^2\,x\,b^2 - 2\,a^2\,b^2 - 2\,a^2 - 2\,b^2 + 1 + b^4 + a^4)^2$$

```
> hsub:=subs(p=-1,h): irreduc(hsub); # The usual triangle, central
angle=2Pi
```

$$false$$

```
> factor(hsub); # The irreducible factors are of degree 3;
# this shows that now the 3-fan is not constructible for cental
angle=2Pi
```

$$(-1 + 2\,x - x^2 + 2\,a^2\,x - 2\,a^2\,x^2 - a^4\,x^2 + 2\,b^2\,x - 2\,b^2\,x^2 - 2\,a^2\,x^2\,b^2 + 4\,a^2\,x^3\,b^2 - b^4\,x^2)^2$$

```
> #
> #Now let us check a lot of rational numbers (about a million)
> num:=1000; den:=1000; sn:=0:#
  for j from 1 to den do
   for i from 1 to j-1 do
     t1:=subs(p= i/j, h): nonconstr:=irreduc(t1);
     if not nonconstr then sn:=sn+1; print(nonconstr,i,j) fi;
     t2:=subs(p=-i/j, h): nonconstr:=irreduc(t1);
     if not nonconstr then sn:=sn+1; print(nonconstr,-i,j) fi;
   od;
  od;   print(sn); # all are irreducible iff sn = 0
```

$$num := 1000$$

$$den := 1000$$

$$0$$

```
> #
> #Now we check a lot of quadratic rationalities p=sqrt(i)/j
> num:=100; den:=100; sn:=0:#
  for j from 1 to den do
   for i from 1 to j^2-1 do
     t1:=subs(p= sqrt(i)/j, h): nonconstr:=irreduc(t1);
     if not nonconstr then sn:=sn+1; print(nonconstr,i,j) fi;
     t2:=subs(p=-sqrt(i)/j, h): nonconstr:=irreduc(t1);
     if not nonconstr then sn:=sn+1; print(nonconstr,-i,j) fi;
```

```
    od;
  od;   print(sn); # all are irreducible iff sn = 0
```

$$num := 100$$

$$den := 100$$

$$0$$

```
> #
> #
> #                Another approach to 3-fans
> # We use sine instead of cosine at the start
>
> fh1:= sin(alpha+beta)^2 - sin(delta-gamma)^2; # this is 0
```

$$fh1 := \sin(\alpha+\beta)^2 - \sin(\delta-\gamma)^2$$

```
> fh2:=expand(fh1);
```

$$fh2 := \sin(\alpha)^2\cos(\beta)^2 + 2\cos(\alpha)\cos(\beta)\sin(\alpha)\sin(\beta) + \cos(\alpha)^2\sin(\beta)^2 - \sin(\delta)^2\cos(\gamma)^2 + 2\cos(\delta)\cos(\gamma)\sin(\delta)\sin(\gamma) - \cos(\delta)^2\sin(\gamma)^2$$

```
> fh3:=subs(sin(alpha)=sqrt(1-cos(alpha)^2),
           sin(beta)=sqrt(1-cos(beta)^2),
           sin(gamma)=sqrt(1-cos(gamma)^2),
           sin(delta)=sqrt(1-cos(delta)^2),fh2);
```

$$fh3 := (1-\cos(\alpha)^2)\cos(\beta)^2 + 2\cos(\alpha)\cos(\beta)\sqrt{1-\cos(\alpha)^2}\sqrt{1-\cos(\beta)^2} + \cos(\alpha)^2(1-\cos(\beta)^2) - (1-\cos(\delta)^2)\cos(\gamma)^2 + 2\cos(\delta)\cos(\gamma)\sqrt{1-\cos(\delta)^2}\sqrt{1-\cos(\gamma)^2} - \cos(\delta)^2(1-\cos(\gamma)^2)$$

```
> fh4:=sort(subs( cos(alpha)=a*u, cos(beta)=b*u,
           cos(gamma)=u, cos(delta)=p, fh3)); # this is still 0
```

$$fh4 := (-b^2u^2+1)a^2u^2 + 2\sqrt{-a^2u^2+1}\sqrt{-b^2u^2+1}\,a\,b\,u^2 + (-a^2u^2+1)b^2u^2 - (-p^2+1)u^2 + 2\sqrt{-p^2+1}\sqrt{-u^2+1}\,u\,p - (-u^2+1)p^2$$

```
> fhaux1:= 2*sqrt(-a^2*u^2+1)*sqrt(-b^2*u^2+1)*a*b*u^2 +
  2*sqrt(-p^2+1)*sqrt(-u^2+1)*p*u;
```

$$fhaux1 := 2\sqrt{-a^2u^2+1}\sqrt{-b^2u^2+1}\,a\,b\,u^2 + 2\sqrt{-p^2+1}\sqrt{-u^2+1}\,u\,p$$

```
> fh5:=(fh4-fhaux1)^2 - fhaux1^2; # =0 since fh4=0
```

$$fh5 := ((-b^2u^2+1)a^2u^2 + (-a^2u^2+1)b^2u^2 - (-p^2+1)u^2 - (-u^2+1)p^2)^2 - (2\sqrt{-a^2u^2+1}\sqrt{-b^2u^2+1}\,a\,b\,u^2 + 2\sqrt{-p^2+1}\sqrt{-u^2+1}\,u\,p)^2$$

```
> fh6:=sort(expand(fh5)); # equals 0
```

$$fh6 := -8a^2b^2u^6p^2 + 4a^2b^2u^6 + 4a^2b^2u^4p^2 + a^4u^4 - 2a^2b^2u^4 + 4a^2u^4p^2 + b^4u^4 + 4b^2u^4p^2 - 2a^2u^4 - 2a^2u^2p^2 - 8\sqrt{-a^2u^2+1}\sqrt{-b^2u^2+1}\sqrt{-p^2+1}\sqrt{-u^2+1}\,a\,b\,u^3\,p - 2b^2u^4 - 2b^2u^2p^2 + u^4 - 2u^2p^2 + p^4$$

```
> fhaux2:=
  -8*sqrt(-a^2*u^2+1)*sqrt(-b^2*u^2+1)*sqrt(-p^2+1)*sqrt(-u^2+1)*a
  *p*b*u^3;
```

$$fhaux2 := -8\sqrt{-a^2 u^2+1}\sqrt{-b^2 u^2+1}\sqrt{-p^2+1}\sqrt{-u^2+1}\, a\, b\, u^3\, p$$

```
> fh7:=(fh6-fhaux2)^2 - fhaux2^2; # equals 0 since so does fh6
```

$$fh7 := (-8 a^2 b^2 u^6 p^2 + 4 a^2 b^2 u^6 + 4 a^2 b^2 u^4 p^2 + a^4 u^4 - 2 a^2 b^2 u^4 + 4 a^2 u^4 p^2 + b^4 u^4 + 4 b^2 u^4 p^2 - 2 a^2 u^4 - 2 a^2 u^2 p^2 - 2 b^2 u^4 - 2 b^2 u^2 p^2 + u^4 - 2 u^2 p^2 + p^4)^2 - 64 a^2 u^6 b^2 (-a^2 u^2 + 1)(-b^2 u^2 + 1) p^2 (-p^2 + 1)(-u^2 + 1)$$

```
> fh8:= subs(u=sqrt(x),fh7);
```

$$fh8 := (-8 a^2 x^3 b^2 p^2 + 4 a^2 x^3 b^2 + 4 a^2 b^2 x^2 p^2 + a^4 x^2 - 2 a^2 x^2 b^2 + 4 a^2 x^2 p^2 + b^4 x^2 + 4 b^2 x^2 p^2 - 2 a^2 x^2 - 2 a^2 x p^2 - 2 b^2 x^2 - 2 b^2 x p^2 + x^2 - 2 x p^2 + p^4)^2 - 64 a^2 x^3 b^2 (1 - a^2 x)(1 - b^2 x) p^2 (-p^2 + 1)(1 - x)$$

```
> fh9:= sort(collect(expand(fh8),x),x);
```

$$fh9 := 16 a^4 b^4 x^6 + (-16 a^2 b^4 - 16 a^6 b^2 p^2 - 16 a^4 b^4 - 16 a^2 b^2 p^2 + 8 a^2 b^6 - 16 a^4 b^2 - 16 b^6 a^2 p^2 + 8 a^2 b^2 + 8 a^6 b^2) x^5 + (1 + 8 a^6 b^2 p^2 + 8 b^6 a^2 p^2 + 16 a^4 b^4 p^4 + b^8 + 8 b^2 p^2 - 4 a^6 b^2 - 4 a^2 b^6 + 4 a^2 b^4 + 4 a^4 b^2 + 4 a^2 b^2 + 8 b^6 p^2 - 16 a^4 b^4 p^2 + 24 a^4 p^2 b^2 + 24 a^2 p^2 b^4 + 6 a^4 b^4 - 16 b^4 p^2 + a^8 + 24 a^2 b^2 p^2 + 16 a^4 p^4 + 16 b^4 p^4 + 8 a^2 p^2 - 16 a^4 p^2 + 8 a^6 p^2 + 6 a^4 + 6 b^4 - 4 a^2 - 4 a^6 - 4 b^2 - 4 b^6) x^4 + (-16 p^6 a^2 b^2 - 16 b^2 p^4 - 4 a^6 p^2 + 4 a^4 p^2 - 16 b^4 p^4 - 4 b^6 p^2 + 4 a^2 p^2 + 24 a^2 p^4 b^2 - 16 a^2 p^4 - 40 a^2 b^2 p^2 + 4 a^4 p^2 b^2 + 4 a^2 p^2 b^4 - 16 b^4 p^4 a^2 + 4 b^4 p^2 + 4 b^2 p^2 - 16 a^4 p^4 - 16 a^4 p^4 b^2 - 4 p^2) x^3 + (4 a^2 p^4 b^2 + 8 b^2 p^6 + 8 p^6 a^2 b^2 + 6 a^4 p^4 + 4 a^2 p^4 + 4 b^2 p^4 + 6 b^4 p^4 + 6 p^4 + 8 a^2 p^6) x^2 + (-4 b^2 p^6 - 4 a^2 p^6 - 4 p^6) x + p^8$$

```
> #
> fh9-h; # This shows that we got the same polynomial!
  #
```

$$0$$

```
> # So the 2nd approach gives nothing new.
>
```